[image: ]

TC2中自动更换PLC运行程序并重启

	摘  要：
为了方便客户升级TC2中的PLC程序，可以借助功能块“FB_CopeFileToFile”实现程序更换。

[bookmark: _Toc109827063]软硬件版本
[bookmark: _Toc109827064]倍福Beckhoff
[bookmark: _Toc109827065]控制器硬件
CX8090
EL2008
EL2004
[bookmark: _Toc109827066]控制软件
笔记本和控制器都是基于TwinCAT 2.11 Build 2307版本
[bookmark: _Toc109827067]TC2中Boot文件夹介绍
在倍福控制器中， TwinCAT 2的项目文件总是位于“hard disk:\TwinCAT\Boot”或“C:\TwinCAT\Boot”中，所以对于某些控制器无法启动的情况，可以通过清空boot文件夹来实现TwinCAT 项目恢复出厂设置，以使控制器恢复至正常状态。
对于TwinCAT2的控制器，其Boot文件夹中的内容一般如下
[image: ]
其中TCPLC_P_1.wbp为引导程序 Boot project，其为每次 TwinCAT PLC Control 中执行 Create BootProject 的时候生成。

因此，对于同一套控制器及其外围设备，如果客户想进行程序升级，可以通过更换TCPLC_P_1.wbp来实现。
[bookmark: _Toc95232430][bookmark: _Toc109827068]FB_CopeFileToFile功能块介绍
[bookmark: _Toc109827069]FB_CopeFileToFile功能块用法
本文档封装了一个自定义功能块“FB_CopeFileToFile”，其功能是把文件夹1中的内容复制至文件夹2中。 

为了实现TCPLC_P_1.wbp的更换升级，则可以把新程序TCPLC_P_1.wbp放置在hard disk:\TwinCAT\Boot_1LED2s文件中，则当bExecute为true时，会覆盖掉hard disk:\TwinCAT\Boot中旧的TCPLC_P_1.wbp。

[bookmark: _Toc95232432][bookmark: _Toc109827070]功能块FB_FileCopy源代码
FUNCTION_BLOCK FB_FileCopy
VAR_INPUT
	srcNetId			: T_AmsNetId;	(* TwinCAT network address of the source file *)
	srcFileName		: T_MaxString;	(* source file path and name *)
	destNetId			: T_AmsNetId;	(* TwinCAT network address of the destination file *)
	destFileName		: T_MaxString;  (* destination file path and name *)
	bExecute			: BOOL;			(* rising edge start fb execution *)
	tAdsTimeOut			: TIME;			(* ads timeout *)
	bLogActive			:			BOOL:=FALSE;
END_VAR
VAR_OUTPUT
	bBusy				:BOOL;
	bError				:BOOL;
	nErrId				:UDINT;
END_VAR
VAR

	fbFileOpen			:FB_FileOpen;
	fbFileClose			:FB_FileClose;
	fbFileRead			:FB_FileRead;
	fbFileWrite			:FB_FileWrite;

	hSrcFile			:UINT	:= 0;(* file handle of the source file *)
	hDestFile			:UINT	:= 0;(* file handle of the destination file *)

	Step				:DWORD;
	RisingEdge			:R_TRIG;

	buffRead			:ARRAY[1..30000] OF BYTE;(* buffer *)
	cbReadLength		:UDINT:=0;
	cbWriteActLength:	UDINT:=0;
	cnRWCounter:		UDINT:=0;

	OnceMore:		BOOL;
	cnRetryCounter:	UDINT;
END_VAR
VAR CONSTANT
	FileIsEmpty:		INT:=16#1234;
END_VAR

RisingEdge(CLK:=bExecute);
IF RisingEdge.Q THEN
	IF NOT bBusy THEN
		bBusy := TRUE;
		bError:= FALSE;
		nErrId:=0;
		Step := 1;
		cbReadLength:=0;
		hSrcFile:=0;
		hDestFile:=0;
		cnRWCounter:=0;
		cnRetryCounter:=0;
	END_IF
END_IF

CASE Step OF
	0:	(* do nothing *)
		;

	1:	(* open source file *)
		OnceMore:=FALSE;
		cnRWCounter:=0;
		fbFileOpen(  bExecute := FALSE  );
		fbFileOpen( 	sNetId := srcNetId,
						sPathName := srcFileName,
						nMode := FOPEN_MODEREAD OR FOPEN_MODEBINARY,
						ePath := PATH_GENERIC,
						tTimeout := tAdsTimeout,
						bExecute := TRUE );
		IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='open source file,step %d' , dintArg := Step);END_IF
		Step := Step + 1;

	2:
		fbFileOpen( bExecute := FALSE );
		IF ( NOT fbFileOpen.bBusy ) THEN
			IF ( fbFileOpen.bError ) THEN
				nErrId := fbFileOpen.nErrId;
				bError := TRUE;
				Step := 50;
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:=CONCAT(CONCAT(srcFileName,destFileName),'File Open Error,ErrorId %d') , dintArg := nErrId);END_IF

			ELSE
				hSrcFile := fbFileOpen.hFile;
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='get source file handle  ,step %d' , dintArg := Step);END_IF

				Step := Step + 1;
			END_IF
		END_IF

	3:	(* open destination file *)
		fbFileOpen(  bExecute := FALSE  );
		fbFileOpen( 	sNetId := destNetId,
						sPathName := destFileName,
						nMode := FOPEN_MODEWRITE OR FOPEN_MODEBINARY,
						ePath := PATH_GENERIC,
						tTimeout := tAdsTimeout,
						bExecute := TRUE );
		IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='open destination file,step %d' , dintArg := Step);END_IF

		Step := Step+1;

	4:
		fbFileOpen(  bExecute := FALSE  );
		IF ( NOT fbFileOpen.bBusy ) THEN

			IF ( fbFileOpen.bError ) THEN
				nErrId := fbFileOpen.nErrId;
				bError := TRUE;
				Step := 50;
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:=CONCAT(CONCAT(srcFileName,destFileName),'File Open Error,ErrorId %d') , dintArg := nErrId);END_IF

			ELSE
				hDestFile := fbFileOpen.hFile;
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='get destination file handle  ,step %d' , dintArg := Step);END_IF

				Step := Step + 1;
			END_IF
		END_IF

	5:	(*read data from source file*)
		cbReadLength := 0;
		fbFileRead( bExecute:= FALSE );
		fbFileRead( 	sNetId:=SrcNetId,
						hFile:=hSrcFile,
						pReadBuff:= ADR(buffRead),
						cbReadLen:= SIZEOF(buffRead),
						bExecute:=TRUE,
						tTimeout:=tAdsTimeOut );
		IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='read data from source file,step %d' , dintArg := Step);END_IF
		
		Step := Step + 1;

	6:
		fbFileRead( bExecute:= FALSE );
		IF ( NOT fbFileRead.bBusy ) THEN
			(*//fbFileRead.bEOF;*)
			IF ( fbFileRead.bError ) THEN
				nErrId := fbFileRead.nErrId;
				bError := TRUE;
				Step := 50;
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:=CONCAT(CONCAT(srcFileName,destFileName),'File Read Error,ErrorId %d') , dintArg := nErrId);END_IF

			ELSE
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='get Read Size  ,step %d' , dintArg := Step);END_IF

				cbReadLength := fbFileRead.cbRead;
				Step := Step + 1;
			END_IF
		END_IF

	7:	(*write data to destination file*)
		fbFileWrite( bExecute := FALSE );
		fbFileWrite(	sNetId:=DestNetId,
						hFile:=hDestFile,
						pWriteBuff:= ADR(buffRead),
						cbWriteLen:= cbReadLength,
						bExecute:=TRUE,
						tTimeout:=tAdsTimeOut );
		IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='write data to destination file  ,step %d' , dintArg := Step);END_IF

		Step := Step + 1;
		cnRWCounter:=cnRWCounter+1;
	8:
		fbFileWrite( bExecute := FALSE );
		IF (  NOT fbFileWrite.bBusy ) THEN
			cbWriteActLength:=fbFileWrite.cbWrite;		
			IF ( fbFileWrite.bError ) THEN
				nErrId := fbFileWrite.nErrId;
				bError := TRUE;
				Step := 50;
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:=CONCAT(CONCAT(srcFileName,destFileName),'File Write Error,ErrorId %d') , dintArg := nErrId);END_IF
			ELSIF (cnRWCounter=1) AND (cbWriteActLength=0) THEN
				
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_ERROR , msgFmtStr:=CONCAT(CONCAT(CONCAT(CONCAT('Read:',UDINT_TO_STRING(cbReadLength)),' write:'),UDINT_TO_STRING(cbWriteActLength)),'_step %d') , dintArg := Step);END_IF
				OnceMore:=TRUE;
				cnRetryCounter:=cnRetryCounter+1;
				IF cnRetryCounter>10 THEN
					bError := TRUE;
					OnceMore:=FALSE;
					nErrId := FileIsEmpty;
				END_IF
				Step := 50; (* repeat reading *)
			ELSE
				
				IF fbFileRead.bEOF THEN (*Check if the EOF flag ist set *)
					Step := 50;		(*cleanup: close the destination and source files *)
					IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='Check if the EOF flag ist set ,step %d' , dintArg := Step);END_IF
					
				ELSE
					Step := 5; (* repeat reading *)
					IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='repeat reading ,step %d' , dintArg := Step);END_IF
					
				END_IF
			END_IF
		END_IF

	30:	(* close the destination file *)
		fbFileClose( bExecute := FALSE );
		fbFileClose( 	sNetId:=destNetId,
						hFile:=hDestFile,
						bExecute:=TRUE,
						tTimeout:=tAdsTimeOut );
		IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='close the destination file ,step %d' , dintArg := Step);END_IF

		Step := Step + 1;

	31:
		fbFileClose( bExecute := FALSE );
		IF ( NOT fbFileClose.bBusy ) THEN
		
			IF ( fbFileClose.bError ) THEN
				nErrId := fbFileClose.nErrId;
				bError := TRUE;
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:=CONCAT(CONCAT(srcFileName,destFileName),'File Close Error,ErrorId %d') , dintArg := nErrId);END_IF

			END_IF
			Step := 50;
			hDestFile := 0;
		END_IF

	40: (* close source file *)
		fbFileClose( bExecute := FALSE );
		fbFileClose( 	sNetId:=srcNetId,
						hFile:=hSrcFile,
						bExecute:=TRUE,
						tTimeout:=tAdsTimeOut );
		IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='close source file ,step %d' , dintArg := Step);END_IF

		Step := Step + 1;

	41:
		fbFileClose( bExecute := FALSE );
		IF (NOT fbFileClose.bBusy ) THEN
		
			IF ( fbFileClose.bError ) THEN
				
				nErrId := fbFileClose.nErrId;
				bError := TRUE;
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:=CONCAT(CONCAT(srcFileName,destFileName),'File Close Error,ErrorId %d') , dintArg := nErrId);END_IF

			END_IF
			Step := 50;
			hSrcFile := 0;
		END_IF

	50: (*error or ready => cleanup*)
		IF ( hDestFile <> 0 ) THEN
			Step := 30; (* close the destination file*)
		ELSIF (hSrcFile <> 0 ) THEN
			Step := 40; (* close the source file *)
		ELSE
			IF OnceMore THEN
				Step:=1;
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='OnceMore! ,step %d' , dintArg := Step);END_IF

			ELSE
				Step := 0;	(* ready *)		
				IF bLogActive THEN ADSLOGDINT(msgCtrlMask:=ADSLOG_MSGTYPE_LOG , msgFmtStr:='Done! ,step %d' , dintArg := Step);END_IF
				bBusy := FALSE;
			END_IF


			
		END_IF

END_CASE
功能块FB_CopeFileToFile源代码
FUNCTION_BLOCK FB_CopeFileToFile
VAR_INPUT
	bExecute:				BOOL;
	sSourcePathName:		STRING:='Hard Disk\TwinCAT\Boot_4LED1s';
	sDestinationPathName:		STRING:='Hard Disk\TwinCAT\boot';
	bLogActive:					BOOL:=FALSE;
END_VAR
VAR_OUTPUT
	bBusy:					BOOL;
	bError:					BOOL;
	bDone:					BOOL;
END_VAR
VAR
	nState:					INT;
	fbCreateDir:				FB_CreateDir;
	bOldExecute:			BOOL;
	nCount:					INT;
	nCount2:				INT;
	fileList:					ARRAY [1..cnMaxNoOfFiles] OF ST_FindFileEntry;
	fbFindFileList: 			FB_EnumFindFileList;
	fbCopyFiles:			FB_FileCopy;

	TonDelay:		TON;
END_VAR
VAR CONSTANT
	cnMaxNoOfFiles:  INT:=50;

END_VAR
CASE nState OF
0:
	IF bExecute AND NOT bBusy THEN
		IF NOT bOldExecute THEN
			bBusy := TRUE;
			bError := FALSE;
			bDone := FALSE;
			nState := 30;
			nCount:=1;
			nCount2:=1;
		END_IF
	END_IF

30:	(* Update file list	*)
	IF nCount <= cnMaxNoOfFiles THEN
		fbFindFileList
			(
			sNetID:='',
			sPathName:=CONCAT(sSourcePathName,'\*.*'),
			eCmd:= eEnumCmd_First,
			pFindList:= ADR(fileList),
			cbFindList:=SIZEOF(fileList),
			tTimeout:=t#10s,
			bExecute := TRUE
			);
		IF fbFindFileList.bExecute AND NOT fbFindFileList.bBusy THEN
			fbFindFileList(bExecute := FALSE);
			fbCopyFiles(srcNetId:='',destNetId:='',	tAdsTimeOut:=t#10s);

			nCount2:=1;

			nState:=40;
		END_IF
		IF fbFindFileList.bError THEN
			nState := 0;
			fbFindFileList(bExecute := FALSE);
			bBusy := FALSE;
			bError := TRUE;
		END_IF
	END_IF
40:
	(* copy each file from source to destination folder	*)
	IF fileList[nCount2].sFileName <> '' THEN
		fbCopyFiles.srcFileName := CONCAT(CONCAT(sSourcePathName,'\'),fileList[nCount2].sFileName);
		fbCopyFiles.destFileName := CONCAT(CONCAT(sDestinationPathName,'\'),fileList[nCount2].sFileName);
		fbCopyFiles(bExecute := TRUE);
		IF fbCopyFiles.bExecute AND NOT fbCopyFiles.bBusy THEN
			fbCopyFiles(bExecute := FALSE);
			nCount2 := nCount2+1;
		END_IF
		IF fbCopyFiles.bError THEN
			bBusy := FALSE;
			bError := TRUE;
			nState := 0;
		END_IF
	ELSE
		nCount:=nCount+1;

			nState:=50;

	END_IF
50:
	TonDelay(in:=TRUE,pt:=t#1s);
	IF TonDelay.Q THEN
		TonDelay(in:=FALSE,pt:=t#1s);
		nState:=0;
		bDone:=TRUE;
		bBusy := FALSE;
	END_IF
END_CASE
[bookmark: _Toc95232434][bookmark: _Toc109827071]测试程序介绍
[bookmark: _Toc109827072]硬件环境
硬件配置为CX8090+EL2008+EL1004，其中EL1004的1-4号针脚关联变量bIn[1..4]，EL2008的1-4号针脚关联变量bIed[1..4]。
[image: 图形用户界面, 文本, 应用程序

描述已自动生成]
[bookmark: _Toc109827073]软件环境
CX8090中原始程序如下，加载了Utilities库。
[image: 图形用户界面, 文本, 应用程序, 电子邮件

描述已自动生成]
现更换程序，把新程序下载到控制器，并create bootproject，然后从boot文件夹中复制TCPLC_P_1到Boot_4LED1s。
[image: 图形用户界面, 文本

描述已自动生成]
再更换程序，把新程序下载到控制器，并create bootproject，然后从boot文件夹中复制TCPLC_P_1到Boot_4LED1s。
[image: 图形用户界面, 文本

描述已自动生成]
分析ACT_changAndReboot程序可知：
[image: 文本

描述已自动生成]
当bIn[3]为true时，nSelect:=1，则新程序的目录为Hard Disk\TwinCAT\Boot_4LED1s，若bStart为true，则换更换新程序并重启。
[image: 图形用户界面

描述已自动生成]
当bIn[3]为false时，nSelect:=2，则新程序的目录为Hard Disk\TwinCAT\Boot_1LED2s，若bStart为true，则换更换新程序并重启。
[image: 图形用户界面

描述已自动生成]

